
Формы, элементы управления  
и события 

5.1. Для чего нуж ны формы 
С самыми простыми возможностями организации взаимодействия с пользо-
вателем (применение встроенных функций MsgBox() и InputBox()) мы уже 
познакомились. Однако, конечно же, возможностей этих функций не всегда 
хватает. В этой главе речь пойдет о том, как создать графический интерфейс 
своего приложения с помощью VBA. 
Чаще всего для предоставления пользователю графического интерфейса ис-
пользуются формы VBA. В принципе, многие элементы управления можно 
вставлять непосредственно на страницу документа (для этого используются 
панели инструментов Формы и Элементы управления), однако классиче-
ский способ — это применение формы. Вне зависимости от того, использует-
ся ли форма или элементы управления размещаются напрямую в документе, 
набор элементов управления и приемы работы с ними одинаковы. 
Как выглядит применение форм в приложении VBA? Обычно форма запуска-
ется при открытии пользователем документа. Пользователь выполняет на 
форме какие-то действия по вводу или выбору информации (например, выби-
рает значения в раскрывающемся списке, устанавливает значения для флаж-
ков и переключателей и т. п.), а потом, как правило, нажимает кнопку на этой 
форме, и введенная им информация передается в базу данных, отправляется 
по электронной почте, записывается в файл для распечатки и т. д. 

5.2. Создание форм  
и самые важ ные свойства и методы форм  
Создать форму очень просто: для этого достаточно в редакторе Visual Basic 
щелкнуть правой кнопкой мыши на проекте (т. е. на имени документа) в окне 

Project Explorer и в контекстном меню выбрать Insert | UserForm. Откроет-
ся окно дизайнера форм (Form designer), в котором будет представлено пус-
тое серое окно формы (по умолчанию она называется UserForm1) и рядом 
Toolbox — панель с набором элементов управления (рис. 5.1). 

 
Рис. 5.1. Все готово для работы с формой 

Если у вас включен показ окна свойств Properties (он включается по клавише 
<F4>), то в этом окне будут представлены свойства формы. Переход к редак-
тору кода для этой формы (по умолчанию открывается событие Click) вы-
полняется по клавише <F7>, возврат обратно в окно дизайнера форм — по 
<Shift>+<F7>. 
Очень удобно, что для форм и элементов управления можно настраивать 
свойства при помощи графического интерфейса окна свойств — резко 
уменьшается количество программного кода, которое нужно писать вручную. 
Некоторые самые важные свойства форм (кроме ShowModal все они примени-
мы и для других элементов управления) приведены далее. 
� Name — это свойство определяет имя формы. Пользователь вашей про-

граммы, скорее всего, его никогда не увидит. Имя формы используется 
только программистом в коде для этой формы (и в окнах редактора Visual 
Basic). После создания формы ее имя, предлагаемое по умолчанию 
(UserForm1), рекомендуется заменить на что-нибудь более значимое, чтобы 

было проще ориентироваться в программе (это относится ко всем элемен-
там управления). 

� Caption — определяет заголовок формы (по умолчанию совпадает с име-
нем формы). Рекомендуется ввести строку, которая будет напоминать 
пользователю о назначении формы (например, "Выбор типа отчета"). 

� Enabled — если это свойство установлено в False, пользователь не сможет 
работать с формой. Используется для временного отключения формы, на-
пример, пока пользователь не обеспечит какие-то условия для ее работы. 

� ShowModal — если свойство установлено в True (по умолчанию), то пользо-
ватель не может перейти к другим формам или вернуться в документ, пока 
не закроет эту форму (так называемый "модальный" режим работы). 

Большая часть других свойств относится к внешнему виду, размерам и ме-
стонахождению формы. 
Самые важные методы форм перечислены в следующем списке. 
� В процессе редактирования формы (из окна редактора Visual Basic) ее 

можно запускать по нажатию клавиши <F5>. После того, как форма будет 
готова, вы должны обеспечить ее запуск в документе. Для запуска формы 
нужно воспользоваться методом Show(): 
UserForm1.Show 

Если форма уже была загружена в память, она просто станет видимой, ес-
ли нет — то будет автоматически загружена (произойдет событие Load). 
Сам этот метод можно вызвать, например: 
• из обычного макроса, привязанного к кнопке или клавиатурной комби-

нации; 
• из автозапускаемого макроса (макроса с названием AutoExec для Word); 
• из кода для элемента управления, расположенного в самом документе 

(например, CommandButton) или на другой форме (для перехода между 
формами); 

• поместить его в обработчик события Open для документа Word или кни-
ги Excel, чтобы форма открывалась автоматически при открытии доку-
мента. 

� После того, как пользователь введет или выберет нужные данные на фор-
ме и нажмет требуемую кнопку, форму необходимо убрать. Для этого 
можно воспользоваться двумя способами: 
• спрятать форму (использовать метод Hide()), например: 

UserForm1.Hide 

Форма будет убрана с экрана, но останется в памяти. Потом при помо-
щи метода Show() можно будет опять ее вызвать в том же состоянии, в 
каком она была на момент "прятанья", а можно, например, пока она 
спрятана, программно изменять ее и расположенные на ней элементы 
управления. Окончательно форма удалится из памяти при закрытии до-
кумента; 

• если форма больше точно не потребуется, можно ее удалить из памяти 
при помощи команды Unload: 
Unload UserForm1 

Остальные методы относятся либо к обмену данными через буфер обмена 
(Copy(), Cut(), Paste()), либо к служебным возможностям формы 
(PrintForm(), Repaint(), Scroll()). 
Важнейшая концепция VBA — события. Событие (event) — это то, что про-
исходит с программой и может быть ею распознано. Например, к событиям 
относятся щелчки мышью, нажатия на клавиши, открытие и закрытие форм, 
перемещение формы по экрану и т. п. VBA построен таким образом, чтобы 
можно было создавать на нем программы, управляемые событиями (event-
driven). Такие программы противопоставляются устаревшему процедурному 
программированию. 
Самые важные события форм приведены далее. 
� Initialize — происходит при подготовке формы к открытию (появлению 

перед пользователем). Обычно в обработчик для этого события помещает-
ся код, связанный с открытием соединений с базой данных, настройкой 
элементов управления на форме, присвоением значений по умолчанию 
и т. п. 

� Click (выбирается по умолчанию) и DblClick — реакция на одиночный и 
двойной щелчок мыши соответственно. Для формы эти события исполь-
зуются не так часто. Обычно обработчики щелчков применяются для кно-
пок (элементов управления CommandButton). 

� Error — это событие используется при возникновении ошибки в форме, 
предоставляя пользователю возможность исправить сделанную им ошиб-
ку. Подробнее — в гл. 6, которая посвящена ошибкам и отладке. 

� Terminate — используется при нормальном завершении работы формы и 
выгрузке ее из памяти (например, по команде Unload). Обычно применяет-
ся для разрыва открытых соединений с базой данных, освобождения ре-
сурсов, протоколирования и т. п. Если работа формы завершается аварий-
но (например, запустившее форму приложение выдало команду End), то 
это событие не возникает. 

Остальные события связаны либо с изменением размера окна формы, либо с 
нажатиями клавиш, либо с активизацией (получением фокуса) или деактиви-
зацией (потерей фокуса). 
Поскольку форма — это во многом просто контейнер для хранения других 
элементов управления, главное ее событие — Initialize. Все остальные со-
бытия обычно используются не для формы, а для расположенных на ней эле-
ментов управления. 
Нужно отметить некоторые моменты, связанные с созданием и редактирова-
нием форм: 
� формы, создаваемые в Microsoft Access, не являются стандартными, как 

формы остальных приложений Office, и набор свойств и методов у них не-
сколько отличается. Тем не менее по функциональности они практически 
одинаковы; 

� иногда для обсуждения форму удобно распечатать. Для этого предусмот-
рено специальное диалоговое окно, которое можно вызвать по нажатию 
клавиш <Ctrl>+<P> (при выбранной форме в дизайнере); 

� если все нужные вам элементы управления трудно уместить на одной 
форме (даже большого размера), в вашем распоряжении два варианта: 
воспользоваться двумя формами (осуществляя переход между ними при 
помощи методов Show() и Hide(), подвязанных к элементам управления) 
или воспользоваться несколькими вкладками для формы. Для этой цели в 
вашем распоряжении — специальный элемент управления Multipage. 

5.3. Элементы управления 

5.3.1. Что такое элемент управления 
Элемент управления — это специализированный объект, который можно 
размещать на формах VBA (или непосредственно в документах) и который 
используется для организации взаимодействия с пользователем. В VBA  
есть как стандартные элементы управления (CommandButton, CheckBox, 
OptionButton), так и нестандартные (любые другие, которые есть на вашем 
компьютере, например, Microsoft Web Browser, представляющий Internet 
Explorer, элемент управления Calendar и т. п.). Элементы управления реаги-
руют на события, которые генерирует пользователь (нажатие на кнопку, ввод 
значения, перемещение ползунка и т. п.). 
Добавление элементов управления на форму чаще всего производится из ди-
зайнера форм при помощи панели Toolbox. Для этого необходимо выбрать 
элемент управления на Toolbox и перетащить его на форму или, что более 

удобно, выделить элемент управления в Toolbox, а затем на форме выделить 
ту область экрана, которую будет занимать этот элемент управления. 
Добавление элементов управления можно производить и программным спо-
собом (при помощи метода Add() коллекции Controls), однако при этом вам 
придется указывать огромное количество свойств создаваемого элемента 
управления, что не очень удобно. 

5.3.2. Элемент управления Label 
Это самый простой элемент управления. Надпись (Label) — это просто об-
ласть формы, в которой выводится какой-то текст (рис. 5.2). 

 

Рис. 5.2. Элемент управления Label на форме 

Пользователь не может изменять этот текст. Чаще всего элемент управления 
Label используется как строка состояния с объяснением того, что сейчас про-
изошло, или происходит, или должен сделать пользователь, и т. п. Этот эле-
мент управления может использоваться и как пояснение для других элемен-
тов управления, таких как ползунок. 
Главное свойство элемента управления Label — это Caption, тот текст, кото-
рый будет выводиться на форме. Большая часть остальных свойств относится 
к форматированию этого текста или настройке внешнего вида этого элемента 
управления. 
Несмотря на то, что для этого элемента управления предусмотрен набор со-
бытий (Click, Error и т. п.), использовать их не принято: пользователю обыч-
но не приходит в голову, что по надписи нужно щелкнуть мышью. 

5.3.3. Элемент управления TextBox 
Текстовое поле (TextBox) — один из самых часто используемых элементов 
управления (рис. 5.3). 

Рис. 5.3. Текстовое поле (элемент управления TextBox) на форме 

Текстовое поле используется: 
� для приема каких-либо текстовых данных, вводимых пользователем (на-

пример, для отправки по почте, для занесения в базу данных и т. п.); 
� для вывода пользователю текстовых данных с возможностью их редакти-

рования (из базы данных, листа Excel и т. п.); 
� для вывода пользователю текстовых данных с возможностью копирования 

и печати, но без возможности изменения (классический пример — текст 
лицензионного соглашения). 

Далее приведены некоторые важные свойства этого элемента управления. 
� Value (или Text, эти два свойства для текстового поля идентичны) — то 

текстовое значение, которое содержится в этом поле. Используется для за-
несения исходного значения и для приема значения, введенного пользова-
телем, в строковую переменную. 

� AutoSize — позволяет текстовому полю автоматически менять свой раз-
мер, чтобы поместить весь текст. Использовать не рекомендуется, т. к. 
может нарушиться весь дизайн вашей формы. 

� ControlSource — ссылка на источник текстовых данных для поля. Может 
ссылаться, например, на ячейку в Excel, на поле в объекте Recordset и т. п. 
При изменении пользователем данных в текстовом поле автоматически 
изменится значение на источнике, определенном в ControlSource. 

� ControlTipText — текст всплывающей подсказки, которая появляется, ко-
гда пользователь наводит указатель мыши на элемент управления. Реко-
мендуется к заполнению для всех элементов управления (для самой фор-
мы это свойство не предусмотрено). 

� Enabled — если установить в False, то текст в поле станет серым и с со-
держимым поля ничего нельзя будет сделать (ни ввести текст, ни выде-

лить, ни удалить). Обычно это свойство используется, чтобы показать 
пользователю, что этот элемент управления отключен до выполнения ка-
ких-либо условий (это относится ко всем элементам управления). 

� Locked — поле будет выглядеть как обычно, пользователь сможет выде-
лять и копировать данные из него, но не изменять их. Используется для 
показа неизменяемых данных типа лицензионных соглашений, сгенериро-
ванных значений и т. п. 

� MaxLength — максимальная длина значения, которое можно ввести в поле. 
Иногда можно использовать свойство AutoTab — при достижении опреде-
ленного количества символов управление автоматически передается дру-
гому элементу управления. 

� MultiLine — определяет, можно ли использовать в текстовом поле не-
сколько строк или только одну. Если вам нужно текстовое поле для прие-
ма одного короткого значения, подумайте, нельзя ли вместо элемента 
управления обойтись функцией InputBox(). 

� PasswordChar — позволяет указать, за каким символом будут "прятаться" 
вводимые пользователем значения. Используется, конечно, при вводе па-
роля. 

� ScrollBars — определяет, будут ли показаны горизонтальная и вертикаль-
ная полосы прокрутки (в любом сочетании). Если текст будет длинным, 
без них не обойтись. 

� WordWrap — настоятельно рекомендуется включать в тех ситуациях, когда 
значение MultiLine установлено в True. В этом случае будет производиться 
автоматический переход на новую строку при достижении границы тек-
стового поля. 

Остальные свойства по большей части относятся к оформлению текстового 
поля и его содержания, а также к настройкам редактирования. 
Главное событие для текстового поля — это событие Change (т. е. изменение 
содержания поля). Обычно на это событие привязывается проверка вводимых 
пользователем значений или синхронизация введенного значения с другими 
элементами управления (например, сделать доступной кнопку, изменить 
текст надписи и т. п.) 

5.3.4. Элемент управления ComboBox 
Комбинированный список (ComboBox) также используется очень часто. Этот 
элемент управления позволяет пользователю выбирать "готовые" значения из 
списка, так и вводить значения самостоятельно (хотя это можно запретить). 
Пример элемента управления ComboBox представлен на рис. 5.4. 

Рис. 5.4. Комбинированный список (элемент управления ComboBox) на форме 

Обычно ComboBox используется в двух ситуациях: 
� когда пользователю необходимо выбрать одно или несколько значений из 

списка размером от 4-х до нескольких десятков позиций. Если позиций 
меньше, то проще использовать переключатель (OptionButton), если боль-
ше — то ориентироваться в списке становится неудобно и необходимо ис-
пользовать специальные приемы, когда пользователь вводит первые буквы 
нужного слова и в списке остаются только значения, которые начинаются 
с этих букв; 

� когда список позиций для выбора необходимо формировать динамически 
на основании данных из источника (базы данных, листа Excel и т. п.). 

К сожалению, через окно свойств заполнить список позициями не получит-
ся — для этой цели придется использовать специальный метод AddItem(). 
Обычно он помещается в обработчик события Initialize для формы. Приме-
нение его может выглядеть так: 

Private Sub UserForm_Initialize() 
    ComboBox1.AddItem "Санкт-Петербург" 
    ComboBox1.AddItem "Ленинградская область" 
    ComboBox1.AddItem "Москва" 
    ComboBox1.AddItem "Московская область" 
End Sub 

Второй параметр varIndex (необязательный) этого метода может использо-
ваться для определения положения элемента в списке, но он не может пре-
вышать значения свойства ListCount и поэтому для начальной загрузки 
ComboBox не подходит. 
Самые важные свойства комбинированного списка представлены далее. 
� ColumnCount, ColumnWidth, BoundColumn, ColumnHeads, RowSource — свойства, 

которые применяются при работе со списками из нескольких столбцов. 
Пользователи не любят такие списки, и поэтому к использованию они  

не рекомендуются (гораздо проще сделать несколько комбинированных  
списков). 

� MatchEntry — определяет, будут ли при вводе пользователем первых сим-
волов значения выбираться подходящие позиции из списка. Возможность 
очень удобная, рекомендуется сохранить значение, которое используется 
по умолчанию. 

� MatchRequired — определяет, разрешается ли пользователю вводить то 
значение, которого нет в списке. По умолчанию False, т. е. разрешено. 

� Value (или Text) — позволяет программным способом установить выбран-
ное значение в списке или получить в переменную выбранное или введен-
ное пользователем значение. 

Остальные свойства (AutoSize, Enabled, Locked, ControlText, ControlTipText, 
MaxLength) применяются точно так же, как и для TextBox. 
Главное событие для комбинированного списка — Change, то же, что и для 
TextBox. Обычно в обработчике этого события проверяются введенные поль-
зователем значения, эти значения переносятся в текстовое поле или в ListBox 
(если нужно дать пользователю возможность выбрать несколько значений, 
поскольку свойства MultiSelect у ComboBox нет) и т. п. 

5.3.5. Элемент управления ListBox 
Элемент управления ListBox очень похож на комбинированный список, но 
применяется гораздо реже по двум причинам: 
� в нем нельзя открыть список значений по кнопке. Все значения видны сра-

зу в поле, аналогичном текстовому, и поэтому большое количество пози-
ций в нем уместить трудно; 

� пользователь не может вводить свои значения — только выбирать из го- 
товых. 

Пример этого элемента управления представлен на рис. 5.5. 
Но у этого элемента управления есть и преимущества: в нем пользователь 
может выбирать не одно значение, как в ComboBox, а несколько. 
Обычно ListBox используется: 
� как промежуточное средство отображения введенных или выбранных 

пользователем через ComboBox значений (или любых других списков, на-
пример, для списка выбранных файлов); 

� как средство редактирования списка значений, сформированных вышеука-
занным образом или полученных из базы данных (для этого можно рядом 
с ListBox разместить кнопки Удалить или Изменить). 

Рис. 5.5. Список (элемент управления ListBox) на форме 

Основные свойства, методы и события у ListBox — те же, что и у ComboBox. 
Главное отличие — это свойство MultiSelect, которое позволяет пользовате-
лю выбирать несколько значений. По умолчанию это свойство отключено. 

5.3.6. Элементы управления CheckBox  
и ToggleButton 
Флажки (CheckBox) (пользователи часто называют их "галками" или "птичка-
ми") и кнопки с фиксацией (ToggleButton) используются для выбора невзаи-
моисключающих вариантов (если этих вариантов немного). Они представле-
ны на рис. 5.6. 

 

Рис. 5.6. Флажки (элементы управления CheckBox) и кнопки с фиксацией (ToggleButton) 

Для CheckBox предусмотрено три главных свойства. 
� Caption — надпись справа от флажка, которая объясняет, что выбирается 

этим флажком. 

� TripleState — если это свойство установлено в False (по умолчанию), то 
флажок может принимать только два состояния: установлен или нет. Если 
для TripleState установить значение True, то появляется третье значение 
Null, когда установлен "серый" флажок. Такое значение часто использует-
ся, например, при выборе компонентов программы при установке, когда 
выбраны не все компоненты, а лишь некоторые. 

� Value — само состояние флажка. Может принимать значения True (флажок 
установлен), False (снят) и Null — "серый" флажок (когда свойство 
TripleState установлено в True). 

Главное событие элемента CheckBox — Change. 
ToggleButton выглядит как кнопка, которая после щелчка на ней остается 
"нажатой" (рис. 5.6), а при повторном щелчке отключается. У нее могут быть 
те же два (или три, в соответствии со свойством TripleState) состояния, что и 
у CheckBox. Свойства и методы — те же самые. Единственное отличие —  
в восприятии их пользователем. Обычно ToggleButton воспринимается поль-
зователем как переход в какой-то режим или начало выполнения продолжи-
тельного действия. 

5.3.7. Элементы управления OptionButton и Frame 
Если CheckBox предназначен для выбора невзаимоисключающих вариантов, 
то переключатель (OptionButton) используется как раз для выбора варианта в 
ситуации "или/или" (рис. 5.7). 

 

Рис. 5.7. Переключатели (2 объекта OptionButton) в рамке (объект Frame) 

Классический пример, при помощи которого можно проиллюстрировать ра-
боту OptionButton, — выбор радиостанции на радиоприемнике: сразу две ра-
диостанции слушать нельзя (поэтому иногда этот элемент управления назы-
вают RadioButton).  

Главных свойств у этого элемента управления два. 
� Caption — надпись для переключателя. 
� Value — установлен переключатель или нет (только два состояния — True 

или False). 
Главное событие тоже стандартное — Change. 
Конечно, использовать один переключатель бессмысленно. Выбор должен 
предоставляться хотя бы из двух вариантов, и при выборе одного из них дру-
гой автоматически снимается. Однако в некоторых ситуациях нам необходи-
мо выбрать из нескольких наборов вариантов (например, отчет за ме-
сяц/квартал/год, тип отчета, нужный филиал и т. п.). Решение простое — пе-
реключатели нужно сгруппировать. 
Самый простой вариант группировки — просто использовать новую форму 
или вкладку на форме. Если переключатели находятся на одной форме (или 
на одной вкладке), они автоматически считаются взаимоисключающими. Ес-
ли же нужно более точно выбрать группы, то необходимо использовать эле-
мент управления Frame. 
Frame — это просто рамка, которая выделяет прямоугольную область на фор-
ме и позволяет организовать элементы управления (рис. 5.7). Помещенные 
внутрь рамки переключатели считаются взаимоисключающими, остальные 
элементы управления ведут себя точно так же, хотя иногда бывает полезно с 
точки зрения наглядности свести вместе под одной рамкой, например, набор 
флажков. При желании рамку можно сделать невидимой, установив для свой-
ства BorderStyle значение 1 и убрав значение свойства Caption. 

5.3.8. Элемент управления CommandButton 
Элемент управления CommandButton (кнопка) — самый распространенный 
элемент управления на формах (рис. 5.8). 

 
Рис. 5.8. Кнопки (объекты CommandButton) 

В большинстве форм обязательно будет, по крайней мере, две кнопки: OK и 
Отмена (Cancel). По нажатию кнопки OK должно выполниться то действие, 

ради чего создавалась форма, по нажатию кнопки Отмена форма должна за-
крыться. Ваша задача — обеспечить необходимый код для этих кнопок, ко-
торый и будет выполнять эти действия. 
Далее представлены самые важные свойства кнопки. 
� Cancel — если для этого свойства установить значение True, то кнопка бу-

дет нажиматься автоматически при нажатии клавиши <Esc>. Как правило, 
на такие кнопки помещаются надписи типа "Отмена", "Выход", "Вернуть-
ся в окно приложения". Однако, кроме назначения клавише <Esc>, свойст-
во ничего больше этой кнопке не дает. Необходимо еще добавить код в 
обработчик события Click, например: 

Private Sub CommandButton1_Click() 
    Unload Me 
End Sub 

Me — это специальное зарезервированное слово, которое представляет те-
кущий объект (в данном случае форму). Его можно использовать вместо 
имени формы. 

� Caption — надпись, которая будет на кнопке. 
� Default — если это свойство установлено в True, то такая кнопка будет 

считаться нажатой при нажатии пользователем клавиши <Enter>, даже ес-
ли фокус находился в другом месте формы (но не на другой кнопке). 
Обычно такие кнопки являются главными, по которым выполняется дей-
ствие, ради которого создавалась форма (печать отчета, занесение инфор-
мации в базу данных, отправка почты и т. п.). 

� Picture — если простая надпись вас не устраивает, можно назначить 
кнопке рисунок (пиктограмму). 

� TakeFocusOnClick — определяет, будет ли передаваться управление этой 
кнопке при нажатии на нее. По умолчанию установлено True. 

Главное событие для кнопки — это, конечно, Click. Как правило, к этому со-
бытию и привязывается программный код, ради которого создавалась кнопка. 

5.3.9. Элементы управления ScrollBar и SpinButton 
Полосы прокрутки (ScrollBars) чаще всего встречаются в текстовых полях, 
когда введенный текст полностью на экране не помещается. Однако ничего 
не мешает вам использовать ScrollBar как отдельный элемент управления 
(пользователи часто называют его "ползунок") для выбора какого-то значения 
из диапазона (рис. 5.9). Обычно такой элемент управления используется для 
выбора плавно меняющихся значений, например: уровня громкости, яркости, 
сжатия, приоритета и т. п. 

Рис. 5.9. Полоса прокрутки (ScrollBar) и счетчик (SpinButton) 

Главное событие для ScrollBar — уже знакомое нам Change. Главные свойст-
ва этого элемента управления представлены далее. 
� Max и Min — максимальное и минимальное значения, которые можно за-

дать при помощи этого элемента управления. Возможный диапазон — 
от −32 767 до +32 767. При этом максимальное значение может быть и 
меньше минимального — просто ползунок придется тянуть в обратную 
сторону. 

� LargeChange и SmallChange — определяют, какими шагами будет двигаться 
ползунок при перемещении его пользователем (путем щелчка на полосе 
около ползунка или при нажатии на одну из кнопок направления соответ-
ственно). 

� Orientation — определяет расположение ползунка (вертикальное или го-
ризонтальное). По умолчанию для этого свойства установлено значение 1, 
т. е. ориентация определяется автоматически в зависимости от конфигура-
ции отведенного элементу управления пространства на форме (что боль-
ше — длина или высота). Однако при помощи этого свойства можно и яв-
но указать вертикальное или горизонтальное расположение ползунка. 

� ProportionalThumb — определяет размер ползунка: будет ли он пропор-
ционален размеру полосы прокрутки (по умолчанию) или будет фиксиро-
ванного размера. 

� Value — главное свойство этого элемента управления. Определяет поло-
жение ползунка и то значение, которое будет возвращать этот элемент 
управления программе. 

Как правило, использование ползунка без отображения выбранной при по-
мощи его информации не очень приветствуется пользователями. В самом 
простом варианте то, что выбрано при помощи ползунка, следует просто ото-
бражать в текстовой надписи: 

Private Sub ScrollBar1_Change() 
    Label1.Caption = ScrollBar1.Value 
End Sub 

В более сложном варианте пользователю можно выбирать — использовать ли 
ползунок или вводить значение в текстовом поле. В этом случае в событии 
Change для текстового поля необходимо предусмотреть проверку вводимых 
пользователем значений и обратную связь с ползунком. 
Элемент управления счетчик (SpinButton) — эта та же полоса прокрутки, 
лишенная самой полосы и ползунка (рис. 5.9). SpinButton используется в тех 
ситуациях, когда диапазон выбираемых значений совсем небольшой (напри-
мер, надо выбрать количество копий для печати отчета). Все свойства, кото-
рые есть у SpinButton, совпадают со свойствами ScrollBar. 

5.3.10. Элементы управления TabStrip и MultiPage 
Набор вкладок (TabStrip) и набор страниц (MultiPage) применяются в одной 
и той же ситуации — когда элементов управления слишком много, чтобы 
уместить их на одной странице формы. Эти элементы управления позволяют 
создавать на форме несколько вкладок (страниц), между которыми сможет 
переходить пользователь. Принципиальное отличие между этими элементами 
управления заключается в том, что на вкладках TabStrip всегда располагают-
ся одинаковые элементы управления, а на MultiPage — разные. Применение 
множества вкладок вы наверняка видели во многих программах (например, в 
Word в окне Параметры, открывающемся с помощью меню Сервис | Пара-
метры). Пример использования элемента управления MultiPage представлен 
на рис. 5.10. 

 

Рис. 5.10. Форма с несколькими вкладками (элементами управления MultiPage) 

Элемент TabStrip используется реже. Например, его можно применить для 
занесения данных по одному шаблону для филиалов или сотрудников (если 
их не слишком много). 

Свойства и события у этих элементов управления практически идентичны. 
Чаще всего используются следующие свойства. 
� MultiRow — определяет, можно ли использовать несколько горизонталь-

ных рядов вкладок. 
� TabOrientation — определяет, где будут расположены заголовки вкладок 

(по умолчанию вверху). 
� Value — номер вкладки, которая открыта в настоящий момент (нумерация 

начинается с 0). 
Главное событие этих элементов управления — Change (т. е. переход между 
вкладками). К нему можно привязать, например, проверку уже введенных 
пользователем значений или вывод предупреждений. 

5.3.11. Элемент управления Image 
Наверное, рисунок (Image) — это самый простой из элементов управления. Он 
позволяет отобразить на форме рисунок в одном из распространенных фор-
матов, который будет реагировать на щелчок мышью (а может просто ис-
пользоваться для украшения формы). Отметим некоторые моменты, связан-
ные с применением элемента управления Image: 
� в качестве альтернативы можно использовать свойство Picture формы 

(особенно если вам нужен фоновый рисунок для всей формы); 
� еще две альтернативы — это свойство Picture элементов управления Label 

или CommandButton. Функциональность рисунков получается практически 
одинаковая; 

� при использовании этого элемента управления само изображение копиру-
ется внутрь документа и внешний его файл больше не нужен. 

Главные свойства этого элемента управления представлены далее. 
� Picture — позволяет выбрать само изображение для формы. 
� PictureAlignment — позволяет выбрать местонахождение изображения в 

отведенной ему области. По умолчанию рисунок располагается по центру. 
� PictureSizeMode — позволяет выбрать режим растяжения или уменьшения 

элемента в случае, если он не соответствует размеру области. 
� PictureTiling — определяет, размножать ли маленький рисунок, чтобы он 

покрыл всю отведенную ему область ("замостить"). 
Главное событие элемента управления Image — Click. 

5.3.12. Применение  
дополнительных элементов управления 
Мы рассмотрели стандартные элементы управления, которые изначально по-
мещены на панель ToolBox и доступны для размещения в формах. Однако 
возможности форм VBA этим не ограничиваются. В вашем распоряжении — 
сотни и тысячи элементов управления, встроенных в Windows, в другие про-
дукты или поставляемые отдельно (в том числе третьими фирмами). Для того 
чтобы можно было разместить их на форме, щелкните правой кнопкой мыши 
по пустому пространству в ToolBox, выберите пункт контекстного меню 
Additional Controls, а затем в списке выберите нужный элемент. Правда, при 
использовании нестандартных элементов управления необходимо помнить, 
что при переносе программы (файла Office) на другой компьютер вам потре-
буется обеспечить на нем наличие необходимых библиотек. 
Очень часто в программах используются дополнительные элементы управле-
ния Internet Explorer, Acrobat Reader, календарь, проигрыватели аудио- и ви-
деофайлов и т. п. Например, чтобы разместить на форме элемент управления 
Microsoft Web Browser (в русифицированной версии Windows он называется 
Обозреватель веб-страниц (Microsoft)), который представляет окно Internet 
Explorer, нужно выполнить следующие действия: 
� щелкнуть правой кнопкой мыши по пустому пространству в окне Toolbox 

и в контекстном меню выбрать Additional Controls; 
� в открывшемся списке выбрать Microsoft Web Browser (или Обозрева-

тель веб-страниц (Microsoft)); 
� изменившимся курсором мыши очертить на форме ту область, которую 

будет занимать этот элемент управления. 
Далее нужно позаботиться о программном коде для этого элемента управле-
ния. Созданный нами на форме элемент управления по умолчанию будет на-
зываться WebBrowser1. Можно выбрать любое из доступных событий этого 
элемента управления, а можно использовать этот элемент управления и в со-
бытиях других объектов. Например, если нам нужно, чтобы при открытии 
формы в окне Internet Explorer на ней открывалась определенная страница, 
можно воспользоваться событием Initialize для формы. Соответствующий 
код может быть таким: 

Private Sub UserForm_Initialize() 
    WebBrowser1.Navigate "http://www.AskIt.ru" 
End Sub 

Преимущества использования этого элемента управления очевидны — вы 
можете расширить функциональность своей формы за счет использования 

Web-страниц (например, с формами HTML). Internet Explorer обычно уста-
новлен на любом компьютере под управлением Windows и поэтому с этим 
элементом управления не возникает никаких проблем при переносе програм-
мы на другой компьютер. Справку по этому элементу управления придется 
смотреть в MSDN. 
Еще один часто используемый элемент управления, который есть практиче-
ски на всех компьютерах — Calendar (календарь) (рис. 5.11). В зависимости 
от версии вашей операционной системы и установленного программного 
обеспечения он может называться по-разному, у меня на компьютере он на-
зывается Calendar Control 8.0. При помощи этого элемента управления 
пользователю будет очень удобно выбирать нужную дату. 

 

Рис. 5.11. Элемент управления Calendar и надпись, дублирующая значение,  
выбранное пользователем в Calendar 

Главное свойство этого элемента управления — Value, т. е. та дата, которая 
выбрана пользователем. Остальные свойства предназначены для отображения 
внешнего вида календаря. 
В Excel на панели ToolBox имеется еще один специфический элемент управ-
ления — RefEdit (в списке Additional Controls он называется как 
RefEdit.Ctrl). Он похож на текстовое поле с кнопкой справа. При нажатии на 
эту кнопку форма, на которой размещен этот элемент управления, "спрячет-
ся", а пользователю будет предоставлена возможность выбрать одну ячейку 
или диапазон ячеек Excel. После того как пользователь завершит выбор, он 
опять вернется в окно формы, а в RefEdit будет помещена информация об 
адресе выбранного диапазона. Такой же адрес, конечно, можно вводить и 
вручную. Главное свойство этого элемента управления — Value. 
Большое количество дополнительных элементов управления предусмотрено 
для форм Access. Они являются специфическими для Access, и про них будет 
рассказано в гл. 12. 

Задание для самостоятельной работы 5: 
Работа с элементами управления 
Подготовка: 
1. Создайте новую книгу Excel и сохраните ее как Prikaz.xls. Заполните ячей-

ки с A1 по A5 значениями, аналогичными представленным на рис. 5.12. 
Данные о сотрудниках лучше ввести в родительном падеже, поскольку эти 
значения будут подставляться в автоматически создаваемый приказ в 
формате документа Word. 

 
Рис. 5.12. Список сотрудников на листе Excel 

2. Откройте редактор Visual Basic и в окне Project Explorer щелкните пра-
вой кнопкой мыши по объекту Эта книга и в контекстном меню выберите 
View Code. 

3. В окне редактора кода для этой книги введите следующий код: 

'При открытии рабочей книги показываем форму UF1 
Private Sub Workbook_Open() 
    UF1.Show 
End Sub 
 
'Специальная процедура, которая печатает приказ в Word 
Public Sub DocWrite(sPovod As String, sFio As String, bFlagPremia As 
Boolean, bFlagGramota As Boolean, nSummaPremii As Long, sOtvIsp As 
String) 
 
    Dim oWord As Word.Application 
    Dim oDoc As Word.Document 
    Set oWord = CreateObject("Word.Application") 

    Set oDoc = oWord.Documents.Add() 
    oWord.Visible = True 
    oDoc.Activate 
    With oWord.Selection 
        .TypeText "Приказ" 
        .Style = "Заголовок 1" 
        .ParagraphFormat.Alignment = wdAlignParagraphCenter 
        .TypeText vbCrLf 
        .Style = "Обычный" 
        .TypeText vbCrLf 
        .TypeText "г.Санкт-Петербург" & Space(90) & Date 
        .TypeText vbCrLf 
        .TypeText vbCrLf 
        .TypeText "За проявленные успехи в " & sPovod & _ 
                  " наградить " & sFio & ":" 
        .TypeText vbCrLf 
        If bFlagPremia Then 
            .TypeText vbTab & "- денежной премией в сумме " & _ 
                      nSummaPremii & " рублей" 
 
        End If 
        If bFlagGramota Then 
            .TypeText vbCrLf 
            .TypeText vbTab & "- почетной грамотой." 
        Else 
            .TypeText "." 
        End If 
        .TypeText vbCrLf 
        .TypeText vbCrLf 
        .TypeText vbCrLf 
        .TypeText vbCrLf 
        .TypeText "Генеральный директор" & vbTab & vbTab & _ 
                  vbTab & "Иванов И. И." 
        .ParagraphFormat.Alignment = wdAlignParagraphCenter 
        .TypeParagraph 
        .TypeText vbCrLf 
        .TypeText vbCrLf 
        .ParagraphFormat.Alignment = wdAlignParagraphLeft 
        .TypeText Text:=("Отв. исполнитель " & sOtvIsp) 
        .TypeParagraph 
 
    End With 
End Sub 

4. В окне Project Explorer щелкните правой кнопкой мыши по проекту 
Prikaz.xls и в контекстном меню выберите Insert | UserForm. Выделите 
созданный вами объект формы и нажмите клавишу <F4>. В окне Proper-
ties введите для свойства Name этой формы значение UF1. 
Поместите на форму из Toolbox единственную кнопку — элемент управ-
ления CommandButton1. Установите для этой кнопки значение свойства 
Caption как "Напечатать приказ" (без кавычек) и измените размеры и ме-
стонахождение этой кнопки, чтобы форма выглядела так, как показано на 
рис. 5.13. 

 
Рис. 5.13. Заготовка для формы с единственной кнопкой 

5. Щелкните правой кнопкой мыши по кнопке CommandButton1 на вашей фор-
ме, в контекстном меню выберите View Code и добавьте в код событий-
ной процедуры для события Click этой кнопки следующий код: 

Private Sub CommandButton1_Click() 
    Dim sPovod As String 
    Dim sFio As String 
    Dim bFlagPremia As Boolean 
    Dim bFlagGramota As Boolean 
    Dim nSummaPremii As Long 
    Dim sOtvIsp As String 
 
    'Подставить данные из формы 
    sPovod = "освоении новых информационных технологий" 
    sFio = "Иванова Ивана Ивановича" 

    bFlagPremia = True 
    bFlagGramota = True 
    nSummaPremii = 100000 
    sOtvIsp = "Петрова П. П." 
    'Конец подстановки данных 
 
    Call ЭтаКнига.DocWrite(sPovod, sFio, bFlagPremia, bFlagGramota, _ 
         nSummaPremii, sOtvIsp) 
End Sub 

6. Запустите вашу форму на выполнение и убедитесь, что она работает:  
выводит  с здаваемый документ Word приказ с фиксированными значе-
ниями. 

в о

ЗАДАНИЕ: 
Измените форму таким образом, чтобы вместо присвоения переменным в вы-
деленном комментариями блоке заранее определенных значений пользова-
тель мог выбирать данные при помощи формы. При этом: 
1. Значение переменной sPovod должно выбираться из трех возможных зна-

чений: "освоении новых информационных технологий", "внедрении новых 
программных продуктов" и значение, которое пользователь может ввести 
через текстовое поле. Используйте для этого набор из трех переключате-
лей и текстовое поле (оно должно быть спрятано, если пользователь вы-
брал один из первых двух переключателей). По умолчанию должно под-
ставляться "освоении новых информационных технологий". 

2. Значение переменной sFio должно выбираться пользователем при помощи 
комбинированного списка. В этот комбинированный список должны ав-
томатически помещаться значения из всех непустых ячеек столбца A лис-
та Excel. По умолчанию должно выбираться значение "Иванова Ивана 
Ивановича". 

 
Образец для работы с ячейками столбца можно получить из ответов к преды-
дущим лабораторным работам. 

3. Значения переменных bFlagPremia и bFlagGramota должны устанавливаться 
в зависимости от состояния двух флажков — "Премия" и "Грамота". По 
умолчанию оба флажка должны быть установлены. Если пользователь 
снял оба флажка, то ему должно выводиться предупреждающее сообще-
ние "Не выбрана ни премия, ни почетная грамота!" с отменой вывода до-
кумента. 

4. Пользователь должен иметь возможность задавать значение переменной 
nSummaPremii либо при помощи полосы прокрутки с диапазоном значений 

от 0 руб. до 100 000 руб., либо при помощи текстового поля. Если флажок 
"Премия" снят, то полоса прокрутки и текстовое поле должны быть спря-
таны от пользователя. 
Ход полосы прокрутки (увеличение или уменьшение значения при щелчке 
на кнопках со стрелками) должен быть равен 100 руб. 
По умолчанию размер премии должен быть равен 100 руб. 

5. Поместите на форму еще одну кнопку Отмена. Эта кнопка должна закры-
вать текущую форму и срабатывать при нажатии клавиши <Esc>. 

6. В заголовке формы должно выводиться значение "Формирование приказа 
о выплате премии". 

Общий вид формы может выглядеть, например, так, как представлено на 
рис. 5.14. 

 
Рис. 5.14. Готовая форма 

Ответ к  заданию 5 
К пункту 1 задания (работа с переключателями и текстовым полем): 
1. В окне Project Explorer два раза щелкните мышью по объекту формы UF1. 

Затем в ToolBox щелкните по объекту Label и отведите место этому эле-
менту управления в верхней части формы. Щелкните правой кнопкой мы-
ши по созданному элементу управления Label1 и в контекстном меню вы-
берите Properties. Измените значение свойства Caption на "За что:" и при 
помощи свойства Font подберите подходящий шрифт и его размер. 

2. В ToolBox щелкните по элементу управления OptionButton и отведите на 
форме место этому элементу управления. Повторите эту операцию еще 
два раза. 

3. Откройте свойства первого переключателя. Измените значение свойства 
Name на optOsvoenie, а значение свойства Caption — на "освоение новых ин-
формационных технологий". Для второго переключателя поменяйте значение 
свойства Name на optVnedrenie и свойство Caption — на "внедрение новых 
программных продуктов", для третьего — на optDrugoe и "другое:" соответ-
ственно. 

4. В ToolBox щелкните по элементу управления TextBox и поместите его в 
нужное место формы. Установите для свойства Name этого элемента управ-
ления значение txtDrugoe. 

5. Щелкните правой кнопкой мыши по пустому месту на форме и в контек-
стном меню выберите View Code. В списке событий в верхней части окна 
редактора кода выберите событие Initialize для UserForm и введите для 
него следующий код: 
optOsvoenie.Value = True 
txtDrugoe.Visible = False 

6. Для события Change переключателя optDrugoe введите следующий код: 
If optDrugoe.Value = True Then 
    txtDrugoe.Visible = True 
Else 
    txtDrugoe.Visible = False 
End If 

7. Перейдите к коду события Click для CommandButton1 и вместо строки: 
sPovod = "освоении новых информационных технологий" 

введите следующий код: 
If optOsvoenie.Value = True Then sPovod = _ 
    "освоении новых информационных технологий" 
If optVnedrenie.Value = True Then sPovod = _ 
    "внедрении новых программных продуктов" 
If optDrugoe.Value = True Then sPovod = txtDrugoe.Value 

8. Запустите форму на выполнение, напечатайте приказ и убедитесь, что все 
работает согласно поставленным условиям. 

К пункту 2 задания (работа с комбинированным списком): 
1. Разместите на форме еще один элемент управления Label с надписью "Ко-

го:" и настройте для него шрифт. 

2. Щелкните в Toolbox по элементу управления ComboBox и выделите для не-
го место на форме. Присвойте созданному элементу управления ComboBox 
имя cbFIO. 

3. Откройте код для события Initialize нашей формы UserForm и дополните 
его следующими строками: 

Dim oColumn As Range 
Dim oCell As Range 
Set oColumn = Columns("A") 
For Each oCell In oColumn.Cells 
    If oCell.Value <> "" Then 
        cbFIO.AddItem oCell.Value 
    End If 
Next 
cbFIO.ListIndex = 0 

4. Перейдите к коду события Click для CommandButton1 и вместо строки: 
sFio = "Иванова Ивана Ивановича" 

введите следующий код: 
sFio = cbFIO.Value 

5. Запустите форму на выполнение и убедитесь, что все работает нормально. 

К пункту 3 задания (работа с флажками): 
1. При помощи ToolBox поместите на форму два элемента управления 

CheckBox. Для первого элемента свойству Name присвойте значение 
chPremia и для свойства Caption — значение "Премия", для второго — 
chGramota и "Почетная грамота" соответственно. 

2. Откройте код события Initialize формы UserForm и дополните его сле-
дующими строками: 

chPremia.Value = True 
chGramota.Value = True 

3. Перейдите к коду события Click для CommandButton1 и вместо строк: 

bFlagPremia = True 
bFlagGramota = True 

введите следующий код: 

bFlagPremia = chPremia.Value 
bFlagGramota = chGramota.Value 



Формы, элементы управления и события 125 

If bFlagPremia = False And bFlagGramota = False Then 
    MsgBox "Не выбрана ни премия, ни почетная грамота!" 
    Exit Sub 
End If 

4. Запустите форму на выполнение и убедитесь, что все работает нормально. 

К пункту 4 задания (применение полосы прокрутки и дублирующего тексто-
вого поля): 
1. Поместите на форму еще один элемент управления Label с надписью  

"Сумма премии:". Присвойте его свойству Name значение lblSum. 
2. Поместите рядом текстовое поле и присвойте его свойству Name значение 

txtSum. 
3. Разместите под текстовым полем элемент управления ScrollBar и при-

свойте следующие значения его свойствам: 
• Name — значение sbSum; 
• Min — значение 0; 
• Max — значение 100 000; 
• SmallChange — значение 100. 

4. Для события Change элемента управления sbSum введите следующий код: 
txtSum.Value = sbSum.Value 

5. Для события Change элемента управления txtSum введите следующий код: 
sbSum.Value = CLng(txtSum.Value) 

 
Такой код является потенциально опасным, поскольку не проверяется вводи-
мое пользователем в текстовом поле значение. Если это значение будет не-
возможно преобразовать в числовое или оно окажется больше 100 000, то воз-
никнет ошибка времени выполнения. Как предупреждать появление ошибок и 
перехватывать их, будет рассмотрено в гл. 6. 

6. Для события Initialize нашей формы UserForm добавьте следующий код: 

sbSum.Value = 100 
txtSum.Value = 100 

7. Для события Change элемента управления chPremia добавьте следующий 
код: 

If chPremia.Value = False Then 
    lblSum.Visible = False 

    txtSum.Visible = False 
    sbSum.Visible = False 
Else 
    lblSum.Visible = True 
    txtSum.Visible = True 
    sbSum.Visible = True 
End If 

8. Для кода Click кнопки CommandButton1 вместо кода: 
nSummaPremii = 100000 

впишите код: 
nSummaPremii = sbSum.Value 

9. Запустите форму на выполнение и убедитесь, что все работает нормально. 

К пункту 5 задания (применение кнопки): 
1. Разместите на поле еще одну кнопку и настройте значения ее свойств сле-

дующим образом: 
• Name — значение btnEscape; 
• Caption — значение "Отмена"; 
• Cancel — значение True. 

2. Для события Click этой кнопки поместите код 
Unload Me 

К пункту 6 задания (изменение заголовка формы): 
1. Щелкните правой кнопкой мыши по пустому месту на форме и в контек-

стном меню выберите Properties. 
2. Для свойства Caption настройте значение "Формирование приказа о выплате 

премии". 
3. Запустите форму на выполнение и убедитесь, что приказы печатаются 

правильно. 
 
 
 


